
Jで将棋の対局を実感して楽しむ

西川　利男
　

　今、将棋の世界では、若い藤井聡太四段が大活躍し、すがすがしい話題とな
っている。新聞誌上でも一般の社会面に棋譜が載るほどである[1]。ひと昔前
には、お年寄りは将棋盤に、駒を並べて楽しんだものでる。そこで、現代のパ
ソコンの画面でも、将棋を動かしてみたいものだと、Jでやってみた。
  [1] 読売新聞、2017/6/27（火）朝刊p.30

１．Jによる将棋システムの概要
　まず、最初にJの画面上での将棋の開局直前の駒配置の盤面をお目にかける。
 

し かしなが
ら、 以下に述
べる ように、J
によ る将棋シ
ステ ムの構築
はな かなか容
易なことではなかった。

１．１　JのGridではできない
　将棋は9×9のます目の中で、駒を動かすのであるからと、JのGrid機能

を利用すれば、最初はすぐにもできると思った。しかし、つぎのような点でJ
のGridでは到底無理だと断念した。
　①「金」、「銀」…などの漢字文字を表記しなければならない。
②　さらには、相手方の盤では、上下逆さの文字表示が必要である。
③　つまり、漢字文字の入出力、フォントの難しさが立ちはだかる。

１．２　Jの通常グラフィックスによるシステム
　しかたがないので、Jの通常のグラフィックスで行うことにする。
そのときの問題点を項目ごとに挙げてみる。
まず、将棋盤と各駒の表示は、つぎのように行う。

　①　必要となる漢字文字とその逆さ文字を画像ファイルとして作成する。　
　②　各駒の画像ファイルを指し手となる9×9のます目の位置に貼り付ける。
次に、棋譜の指し手に応じた駒の移動と表示について、ややこしい問題があ

る。
③　新しく指した駒表示は、その位置に画像ファイルを貼り付けるだけなの

で問題はない。
④　同時に移動する以前の位置を空白にすることが必要である。
ここに難しい問題が生じる。
⑤　指した駒はどこから、移動したか。これは駒の種類によって異なる。つ

まり、駒の種類に応じた動きのルールに従って、どこからの移動かを検索しな
ければならない。
結局、Jによるパソコン将棋システムでは、大きくつぎの２つの課題を行う

ことになる。
①　盤面に駒を表示するグラフィックス
②　駒の動きを表現するアルゴリズム

２．Jによる将棋システムのプログラム
２．１ 駒の画像ファイル　
 つぎのように、bmpの小さい画像ファイルを作った。

KI.bmp  
KJ.bmp       GI.bmp       GJ.bmp       HU.bmp     HUN.bmp
 

 …　…
 
 
 OH.bmp        HS.bmp       KA.bmp

２．２　将棋盤面のJグラフィックスによるフォームの作成
　通常のように、Jのグラフィックス gl2ライブラリを用いるウィンドウズ・
プログラムとして、画像isigraph の ShBan、棋譜を進めるボタン Start, Go、
指し手表示のためのエディットボタン Sashiteなどを持つフォームを作成した。

require 'gl2'

SHOGI=: 0 : 0
pc shogi;
menupop "File";
menu new "&New" "" "" "";
menu open "&Open" "" "" "";
menusep ;
menu exit "&Exit" "" "" "";
menupopz;
xywh 191 37 34 11;cc Go button;
xywh 190 18 34 11;cc Start button;
xywh 192 163 34 12;cc ok button;cn "Test";
xywh 191 182 34 12;cc cancel button;cn "Exit";
xywh 3 17 181 180;cc ShBan isigraph;
xywh 18 3 45 11;cc Sashite edit ws_border es_autohscroll;
pas 6 6;pcenter;
rem form end;
)

Path =: 2{. 1!:40 ''

PathShBan =: Path ,'\j402\user\Shogi_Koma\'

NB. run '' => no trace
NB. run 10 => trace after 10 
run =: shogi_run
shogi_run=: 3 : 0
if. y. = ''
  do. TestSt =: 0
  else. TestSt =: ". y.
end. 
wd SHOGI
NB. initialize form here

St =: 0
wd 'pshow;'
)　

２．３ 将棋盤の作成と駒の画像ファイルの貼り付け
　ボタン Startを押して始める初期画面プログラム
shogi_Start_button=: 3 : 0
NB. 9 x 9 draw lines for Shogi_ban ========================
gllines 40 120 940 120
gllines 40 120 40 940
gllines 140 120 140 940
gllines 240 120 240 940
gllines 340 120 340 940
gllines 440 120 440 940
gllines 540 120 540 940
gllines 640 120 640 940
gllines 740 120 740 940
gllines 840 120 840 940
gllines 940 120 940 940
gllines 40 940 950 940
gllines 40 220 940 220
gllines 40 310 940 310
gllines 40 400 940 400
gllines 40 490 940 490
gllines 40 580 940 580
gllines 40 670 940 670
gllines 40 760 940 760
gllines 40 850 940 850
NB. paint koma figures on start positions ===========================
j =. 0
while. j < 9
   do.
   i =. 0
   while. i < 9
      do. 
         koma_display j, i
         i =. i + 1
   end.
   j =. j + 1
end.
glshow ''
)
左引数、盤位置(x, y) に、右引数、駒ファイル(例 KI.bmp)を貼り付ける 
koma_display =: 3 : 0
x =. 0{y.
y =. 1{y.
glbmpxywh (50 + 100 * y), (850 - 90 * x), 80, 80
glbmp PathShBan, (>(< x, y) { ShBan), '.bmp'
)   　

２．４ Jの内部処理への対応
　駒の移動の処理のために、システムの内部では変化する盤面を記録するJの
9 x 9の配列が必要である。これを配列ShBanとし英字による駒名をきめる。
  開始の盤面ではつぎのようになる。
   ShBan
+--+--+--+--+--+--+--+--+--+
|KZ|KF|GJ|KJ|OI|KJ|GJ|KF|KZ|
+--+--+--+--+--+--+--+--+--+
|_ |HT|_ |_ |_ |_ |_ |KB|_ |
+--+--+--+--+--+--+--+--+--+
|HV|HV|HV|HV|HV|HV|HV|HV|HV|
+--+--+--+--+--+--+--+--+--+
|_ |_ |_ |_ |_ |_ |_ |_ |_ |
+--+--+--+--+--+--+--+--+--+
|_ |_ |_ |_ |_ |_ |_ |_ |_ |
+--+--+--+--+--+--+--+--+--+
|_ |_ |_ |_ |_ |_ |_ |_ |_ |
+--+--+--+--+--+--+--+--+--+
|HU|HU|HU|HU|HU|HU|HU|HU|HU|
+--+--+--+--+--+--+--+--+--+
|_ |KA|_ |_ |_ |_ |_ |HS|_ |
+--+--+--+--+--+--+--+--+--+
|KY|KE|GI|KI|OH|KI|GI|KE|KY|
+--+--+--+--+--+--+--+--+--+
　
　なお、グラフィック画面での盤面は単に表示だけある。すべての駒の操作は
この配列ShBanに対しておこなう。
　伝統的な将棋の棋譜では、「２六歩」のように右上を開始のます目として、
ヨコタテで表示されるが、これらをJの配列（左上からi=0,,8, j=0,,8 ）で
処理するための文字、インデックスの変換が必要である。これらの変換プログ
ラムはつぎのようになる。
KN =: '１';'２';'３';'４';'５';'６';'７';'８';'９'
KM =: '一';'二';'三';'四';'五';'六';'七';'八';'九'
kanj_to_n =: 3 : 0
RG =. y.
R =. (0, 1){RG
r =. >: (> */ L:0 R = L:0 KN) # i.9
G =. (2, 3){RG
g =. >: (> */ L:0 G = L:0 KM) # i.9
r, g
)

ind_adj =: 3 : 0
(_1 + (1{y.)), (9 - (0{y.))
) 

NB. Kanji to i, j number  Example Usage: ==========
NB.   Kan_ind =: , > }: 0 {SDATA
NB.   Kan_ind
NB. ２六
NB.   ind_adj kanj_to_n  Kan_ind
NB. 5 7
NB.    
NB.    Kan_ind =: , > }: 1 {SDATA

NB.    Kan_ind
NB. ８四
NB.   ind_adj kanj_to_n  Kan_ind
NB. 3 1
NB. ===============================================

KOMAJ =: '歩';'香';'桂';'銀';'金';'王';'飛';'角'
KOMAA =: 'HU';'KY';'KE';'GI';'KI';'OH';'HS';'KA'
KOMAB =: 'HV';'KZ';'KF';'GJ';'KJ';'OI';'HT';'KB'
 
２．５ 指し手に応じた駒の移動のアルゴリズム 
　これからが、ややこしいところである。例として「銀」の駒を動かすときを、
説明しよう。
　棋譜データから、順次読み込まれた駒の指し手は、「６八銀」となっている、
とする。偶数番目は先手である。上の変換で、これは「7 3 GI」となり、これ
をJで処理する。まず、配列ShBanの (7, 3)の要素の値として、文字「GI」に
置き換える。
　つぎに、ここで駒「銀」が動いて来た元のます目の位置を検索し、そのます
目を空白にしなければならない。そこで、駒「銀」の動きのルールからこの検
索を行う。
 「銀」のます目の位置、配列ShBanの(7, 3)に対して、「銀」の動きのルール
から、移動元の可能性の位置は
　　 (8 2), (8 3), (8 4), (6 2), (6 4)
となる。したがって、配列ShBanからこれらの要素である文字を得ると以下に
ようになる。
       GI,   _ ,     OH,    KA,    HU
つまり、最初に出会ったGI に等しいインデクス(8 2)が元の位置である。この
位置の文字を空白文字に置き換えればよいのである。
　このようにして、駒の移動が行われる。

２．６ 指し手に応じた駒の移動処理などのJプログラム　
　　ボタンGoを押して行う指し手のJ処理プログラムは以下のようになる。 
NB. Sashite Display one by one with press down Go button ===============
shogi_Go_button=: 3 : 0 
wr 'No. ', (": St), ' ======='
if. 91 = St do. wr '=== end ===' goto_utikoma. end.
Sab =. St { SDATA
UT =. shogi_Test_button St  NB. 付加文字
NB. display Sashite data on edit box
SS =. (> 0{Sab), (> 1{Sab), (> 2{Sab)
wr SS,'  ', (": St),' ', >(2|St){('先手';'後手')
wd 'set Sashite * ', SS,'  ', (": St),' ', >(2|St){('先手';'後手')
Kan_ind =: , > }: SS
'i j' =: ind_adj kanj_to_n Kan_ind
Komaj =: (4, 5) {, > SS
SG =. (2|St){(1, _1)                     NB. 先手:SG=+1, 後手:SG=_1
Komaa =. SG sgsel Komaj
wr (": i, j),' ', , Komaa
(, Komaa) koma_paint i, j               NB. 指手の駒名のグラフ盤面表示
ShBan =: (<, Komaa) (<i, j) } ShBan     NB. 棋譜配列更新、駒名
if. ((St >: TestStA) *. (St <: TestStB)) do.  wr ShBan end. 
if. 1 = UT do. goto_utikoma. end.
select. Komaj   NB. 駒の動きの選択
  case. '歩' do.
    ij =. (i+SG), j                 
  case. '金';'と' do. 

  ij =. ((i+SG),j-1);((i+SG),j);((i+SG),j+1);(i,(j-1));(i,(j+1));((i-SG),j)
  case. '銀' do.
    ij =. ((i+SG),j-1);((i+SG),j);((i+SG),j+1);((i-SG),j-1);((i-SG),j+1)
  case. '桂' do.
    ij =.  ((i+2*SG),j-1);((i+2*SG),j+1)
  case. '王' do.
    ij1 =. ((i+SG),j-1);((i+SG),j);((i+SG),j+1);((i-SG),j-1)
    ij2 =. (i,j-1);(i,j+1)
    ij3 =. ((i-SG),j);((i-SG),j+1)
    ij =. ij1, ij2, ij3  
  case. '飛';'竜' do.
    ija =. }. <"(1) (9 | i + i.9) ,. j     
    ijb =. }. <"(1) i ,. (9 | j + i.9)  
    ij =. ija, ijb
  case. '角';'馬' do. 
    ijc =. }. <"(1) (9 | i + i.9) ,. (9 | j + i.9)  
    ijd =. }. <"(1) (9 | i - i.9) ,. (9 | j + i.9)  
    ij =. ijc, ijd
end.
NB. 空白の検出 =====================================================
    ij =. 9 |(L:0) ij                       NB. 候補インデックス、調整
    AA =. ij { ShBan                        NB. 候補駒名の取り出し
    BB =. (<,Komaa) = ij { ShBan            NB. 同一テストをした後、
                                            NB. 単一インデックスの取り出し

 if. ((St >: TestStA) *. (St <: TestStB)) do. 
      wr ij
      wr AA
      wr BB
end. 
select. Komaj
  case. '金';'銀';'王';'桂';'飛';'角';'と';'竜';'馬' do.
    ij =. , > BB # ij
    ij =. 9 | ij
  end.
NB. '駒名上' の処理 ===============
  if. 3 = UT
    do.
      ij =. (0, 1) { ij
    end.
NB. '駒名直' の処理 ===============
  if. 4 = UT
    do.
      ij =. (2, 3) { ij
    end.

NB. '駒名成' の処理 ===============
  if. 2 = UT
    do.
      KOMAN =. (, Komaa),'N'  NB. rename komaa, attaching 'N'  
      ShBan =: (<, KOMAN) (<i, j) } ShBan     NB. 棋譜配列更新、駒名  
      (KOMAN) koma_paint i, j                 NB. 指手の駒名のグラフ盤面
    end.
NB. 空白処理後表示 ================
  if. ((St >: TestStA) *. (St <: TestStB)) do. 
      wr (": ij), 'BL'
  end.
  ('BL') koma_paint ij                 NB. 駒の移動後の空白のグラフ盤面表示
   ShBan =: (<'_') (<ij) } ShBan       NB. 棋譜配列更新、空白
   if. ((St >: TestStA) *. (St <: TestStB)) 

     do. wr ShBan
   end. 
label_utikoma.
St =: St + 1
)
　

２．７ 指し手に応じた駒の移動のJの実行例

・指し手「６八銀」からJの表示に変換して
　「銀」の場合を選択して(select. case. 構文)から、
ShBan の(7 3)の要素をGIにする

６八銀  8 先手
7 3 GI
+--+--+--+--+--+--+--+--+--+
|KZ|KF|GJ|KJ|OI|_ |GJ|KF|KZ|
+--+--+--+--+--+--+--+--+--+
|_ |HT|_ |_ |_ |_ |KJ|KB|_ |
+--+--+--+--+--+--+--+--+--+
|HV|_ |HV|HV|HV|HV|_ |HV|HV|
+--+--+--+--+--+--+--+--+--+
|_ |_ |_ |_ |_ |_ |HV|_ |_ |
+--+--+--+--+--+--+--+--+--+
|_ |HV|_ |_ |_ |_ |_ |_ |_ |
+--+--+--+--+--+--+--+--+--+
|_ |_ |HU|_ |_ |_ |_ |HU|_ |
+--+--+--+--+--+--+--+--+--+
|HU|HU|KA|HU|HU|HU|HU|_ |HU|
+--+--+--+--+--+--+--+--+--+
|_ |_ |KI|GI|_ |_ |_ |HS|_ |
+--+--+--+--+--+--+--+--+--+
|KY|KE|GI|_ |OH|KI|GI|KE|KY|
+--+--+--+--+--+--+--+--+--+

・GI の動きのルールから可能な移動元を検索し、
空白にすべきインデックス(8 2)を得て、その位置の値を空白(_)にする

+---+---+---+---+---+
|8 2|8 3|8 4|6 2|6 4|
+---+---+---+---+---+
+--+-+--+--+--+
|GI|_|OH|KA|HU|
+--+-+--+--+--+
1 0 0 0 0
8 2 BL

+--+--+--+--+--+--+--+--+--+
|KZ|KF|GJ|KJ|OI|_ |GJ|KF|KZ|
+--+--+--+--+--+--+--+--+--+
|_ |HT|_ |_ |_ |_ |KJ|KB|_ |
+--+--+--+--+--+--+--+--+--+
|HV|_ |HV|HV|HV|HV|_ |HV|HV|
+--+--+--+--+--+--+--+--+--+

|_ |_ |_ |_ |_ |_ |HV|_ |_ |
+--+--+--+--+--+--+--+--+--+
|_ |HV|_ |_ |_ |_ |_ |_ |_ |
+--+--+--+--+--+--+--+--+--+
|_ |_ |HU|_ |_ |_ |_ |HU|_ |
+--+--+--+--+--+--+--+--+--+
|HU|HU|KA|HU|HU|HU|HU|_ |HU|
+--+--+--+--+--+--+--+--+--+
|_ |_ |KI|GI|_ |_ |_ |HS|_ |
+--+--+--+--+--+--+--+--+--+
|KY|KE|_ |_ |OH|KI|GI|KE|KY|
+--+--+--+--+--+--+--+--+--+
   
２．８ 指し手棋譜データベース
　藤井四段と益田四段の最年少記録がかかった対局の棋譜を格納してある。
NB. Shogi Go Routine ========================================
NB. Sashite Display on edit box Sashite =====================
NB. Sashite_Jun Data
S1 =: (1,3)$'２';'六';'歩'
S2 =: (1,3)$'８';'四';'歩'
S3 =: (1,3)$'７';'六';'歩'
S4 =: (1,3)$'３';'二';'金'
S5 =: (1,3)$'７';'八';'金'
S6 =: (1,3)$'８';'五';'歩'
S7 =: (1,3)$'７';'七';'角'
S8 =: (1,3)$'３';'四';'歩'
S9 =: (1,3)$'６';'八';'銀'
S10 =: (1,3)$'４';'四';'歩'

　（以下省略）

SDATA =: S1, S2, S3, S4, S5, S6, S7, S8, S9, S10, ...

３．Jによる将棋システムの実際
　中盤からの戦闘の開始、後手８六歩で突いてきたのを先手同歩で受けた。

ところが、後手は思い がけな

いところから角を繰り出 しこれ
を取った。
　
　

　
　
　
　
　
　
　
　

そ れに対して先手は同角と
受 けた。

　さらにこれに対して、後手は同飛車と、両者の息のつまるような応戦の盤面
のようすをみていただきたい。　
　
　
　
　
　
　
　
　
　
　
　
　
　
 

  
 
 
 
 
 
 
 　
　
　なお、 これらの実
行画面 はJAPLA 

2017/8/5例
会で発 表時に、鳥
邊錬太 郎氏、
志村正 人氏より、
ご意見ご指摘を受けて、以下のようなシステムへの追加、改良を行ったもので
ある。
　(1) 現在の指し手を、青色の円で囲んでその位置を示す。
　(2) 先手、後手のそれぞれについて、現在の持ち駒を表示する。
　そのためのプログラムの手直しは、かなりなもので、すっかりJプログラミ
ングを楽しませてくれた。
　

　

  
 
 
 　

　その後、後手による２七角打ちという手により、飛車、金取りというつぎの
ような局面もあった。　

 ４． 最後に－将
棋と 音楽
　Jで 夏の気楽な
話題 にと思って、
始め てみたもの
の、 これほどの
仕事になるとは、思ってもみなかった。
　将棋そのものも、子どもの頃にやって以来、小学生の孫とやったくらいだ。
その孫たちも大学を卒業してしまって、最近はよりつかない。このあいだ、つ
れづれなるままに、ワイフと２，３回指したことがあったけ、というところだ。
　いつ買ったのか、大山康晴名人の「将棋の勝ち方」などという本があった。
将棋は棋譜をただ眺めて目で追っているだけでは、まさに絵にかいた餅だ。
　今回、Jの上でボタンを押して、実際に駒を動かしてみて、思いがけなく、
心躍り、相手との切ったはったの緊張感を感じた。
　ちょうど楽譜の上の音符は、それを実際に演奏したり歌ったりして、初めて
いのちがかよってくる、というの同じ体験であった。

